/*
* Copyright 2011 Pavel Kozhin 
*
* Licensed under the Apache License, Version 2.0 (the "License"); 
* you may not use this file except in compliance with the License. 
* You may obtain a copy of the License at 
* 
* http://www.apache.org/licenses/LICENSE-2.0
* 
* Unless required by applicable law or agreed to in writing, software 
* distributed  under the License is distributed on an "AS IS" BASIS,
* WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied. 
* See the License for the specific language governing permissions and 
* limitations under the License.
*
* Author: Pavel Kozhin.
*/
package org.etherframework.mediation.core
{
	import org.etherframework.IEther;
	import org.etherframework.common.mediation.IMediationBind;
	import org.etherframework.mediation.manager.IMediationManager;
	
	[ExcludeClass]
	
	public final class MediationBind implements IMediationBind
	{
		private var _manager:IMediationManager;
		private var _isBind:Boolean;
		private var _ether:IEther;
		private var _info:MediationData;
		
		public function MediationBind(manager:IMediationManager)
		{
			_manager = manager;
		}
		
		public function prepare(ether:IEther, info:MediationData, isBind:Boolean):IMediationBind
		{
			_ether = ether;
			_info = info;
			_isBind = isBind;
			return this;
		}
		
		public function method(term:*, handler:Function):void
		{
			if (!(handler is Function))
				throw new ArgumentError("Argument handler:Function can not be null.");
			
			if (_isBind)
			{
				_manager.bindMethodAsTerm(_ether, handler); 
			}
			else
			{
				_manager.unbindMethodAsTerm(_ether, handler);
			}
			
			clear();
		}
		
		public function property(term:*, setter:Function, getter:Function=null):void
		{
			if (setter is Function || getter is Function)
			{
				(_isBind) ? _manager.bindProperty(_ether, setter, getter) :
					_manager.unbindProperty(_ether, setter, getter);
			}
			
			clear();
		}
		
		private function clear():void
		{
			_info.clear();
			_info = null;
			_ether = null;
		}
	}
}